Introduction to Computational Algebraic Geometry

Jan Verschelde

University of Illinois at Chicago
Department of Mathematics, Statistics, and Computer Science
`http://www.math.uic.edu/~jan`
`jan@math.uic.edu`

UIC ASCEND Workshops, 24 July 2008

Computational Algebraic Geometry

an introduction to a modern mathematical discipline

The big picture:

- What is algebraic geometry?

  Algebraic geometry studies solutions of polynomial systems.

  Polynomial systems occur in a wide variety of applications.

- Using computers to discover theorems.

  Computer algebra software offers implementations of algorithms to solve polynomial systems.

  We will use SAGE, an open source software system.

Problem of today:

How do two circles intersect?

Outline

1. SAGE: Software for Algebra and Geometry Experimentation
   - Try it online!

2. An Intersection Problem
   - plotting and solving specific instances
   - looking at the general problem formulation

3. Determinants, Resultants and Discriminants
   - Jacobian matrices and singular solutions
   - eliminating variables with resultants
   - computing discriminants using resultants

SAGE is open source mathematical software

1. compilation both of original Python, C, C++, and SageX code
2. interfaces to computational algebraic geometry software: Singular
3. the GUI is your web browser, try it before installation

Three steps to getting started:

1. Go to `http://www.sagemath.org`
2. click on **Try it online!**
3. Sign up for a new SAGE notebook account.

Plotting Two Circles

Consider two circles, how do they intersect?




```
sage: c1 = circle( (1,2) , 3 , rgbcolor=(1,0,0) )
sage: c2 = circle( (-1,1) , 2 , rgbcolor=(0,0,1) )
sage: c12 = c1 + c2
sage: c12.show(aspect_ratio=1)
```

Computing the Intersection Points

algebraic problem formulation: solve a polynomial system

We solve a system of two polynomial equations in two unknowns:

```
sage: x,y = var('x, y')
sage: p1 = (x-1)^2 + (y-2)^2 - 3^2 == 0
sage: p2 = (x+1)^2 + (y-1)^2 - 2^2 == 0
sage: sols = solve([p1,p2],x,y)
sage: sols
```

We obtain two solutions in symbolic form:

```
[[x == (-2*sqrt(5) - 5)/5, y == (4*sqrt(5) + 5)/5],
 [x == (2*sqrt(5) - 5)/5, y == (5 - 4*sqrt(5))/5]]
```

Verifying the Solutions

```
sage: print sols[1]
sage: vx = sols[1][0].rhs()
sage: print n(vx,200)
sage: vy = sols[1][1].rhs()
sage: s = p1.substitute(x=vx,y=vy)
sage: print s
sage: s.expand()
```

```
[x == (2*sqrt(5) - 5)/5, y == (5 - 4*sqrt(5))/5]
-0.105572809000084121436330532507489505823752656155389
  2 sqrt(5) - 5     2    5 - 4 sqrt(5)     2
 (------------- - 1)  + (------------- - 2)  - 9 == 0
        5                      5
0 == 0
```

Likewise we do it for the second solution and also for p2.

Choice of Coordinate System

With out loss of generality we may choose

- the origin is center of the first circle
- the radius of the first circle to be one

$\rightarrow$ first circle is the unit circle:

$$f = x^2 + y^2 - 1$$

We may choose the orientation of the x-axes

- through the center of the second circle: $(c, 0)$
- let r be the radius of the second circle

$\rightarrow$ two parameters for the second circle:

$$g = (x - c)^2 + y^2 - r^2$$

Our problem is governed by two parameters: c and r.

A General Solution

symbolic computation manipulates symbols as numbers

We declare c and r as variables and solve:

```
sage: c,r = var('c, r')
sage: f = x^2 + y^2 - 1
sage: g = (x-c)^2 + y^2 - r^2
sage: solve([f==0,g==0],x,y)
```

We obtain a symbolic solution:

$$\left[x = \frac{-r^2 + c^2 + 1}{2c}, y = \pm \frac{\sqrt{-r^4 + 2c^2r^2 + 2r^2 - c^4 + 2c^2 - 1}}{2c} \right]$$

but how general is this solution?

Singular Solutions

1. double solutions: two circles touching each other,
2. a solution set: two overlapping circles.

At a singular solution the determinant of the Jacobian matrix vanishes.

The Jacobian matrix collects all partial derivatives:

```
sage: J = matrix([[diff(f,x),diff(f,y)],\
                  [diff(g,x),diff(g,y)]])
sage: print J
sage: dJ = det(J)
sage: print dJ
```

```
[      2*x         2*y]
[2*(x - c)         2*y]
```

```
4 x y - 4 (x - c) y
```

The Determinant

when does a linear system have a singular solution?

Given a linear system $A\mathbf{x} = b$:

$$\left\{ \begin{array}{l} a_{11}x_1 + a_{12}x_2 = b_1 \\ a_{21}x_1 + a_{22}x_2 = b_2 \end{array} \right. \quad \left[\begin{array}{cc} a_{11} & a_{12} \\ a_{21} & a_{22} \end{array} \right] \left[\begin{array}{c} x_1 \\ x_2 \end{array} \right] = \left[\begin{array}{c} b_1 \\ b_2 \end{array} \right].$$

The system $A\mathbf{x} = b$ has a unique solution $\Leftrightarrow \det(A) \neq 0$.

We have explicit formulas to compute a determinant:

$$\det(A) = \det \left[\begin{array}{cc} a_{11} & a_{12} \\ a_{21} & a_{22} \end{array} \right] = a_{11}a_{22} - a_{21}a_{12}.$$

Generalization to more than two equations:

- recursive: expansion along row or column;
- alternative: elimination via row reduction.

The Discriminant

generalizes the determinant to polynomial systems

We will compute the discriminant of the circle problem.

The discriminant is a polynomial in c and r which will vanish whenever the solutions to the circle problem are singular.

Adding the determinant of the Jacobian matrix to the system, we solve

$$\left\{ \begin{array}{r} x^2 + y^2 - 1 = 0 \\ (x - c)^2 + y^2 - r^2 = 0 \\ 4cy = 0 \end{array} \right.$$

Looking long enough at the system will lead to the solutions...

Our aim is to illustrate a general approach.

Definition of the Resultant

a tool to solve polynomial systems

Given two polynomials p and q with general coefficients:

$$p(x) = a_2x^2 + a_1x + a_0$$
$$q(x) = b_2x^2 + b_1x + b_0$$

For which values of the coefficients do p and q have a common factor?

The resultant

- is a polynomial in the coefficients of p and q
- is zero for those coefficients for which p and q have a common factor

Application: eliminate x.

Resultants to Eliminate Variables

Suppose p and q do have a factor f: $p = Pf$, $q = Qf$.

Observe: $Qp = QPf$ and $Pq = PQf$ imply $Qp = Pq$.

$$\begin{array}{ll} p(x) = a_2x^2 + a_1x + a_0 & P(x) = \alpha_1 x + \alpha_0 \\ q(x) = b_2x^2 + b_1x + b_0 & Q(x) = \beta_1 x + \beta_0 \end{array}$$

Elaborate the condition $Qp = Pq$ and consider

$$(\beta_1 + \beta_0)(a_2x^2 + a_1x + a_0) = (\alpha_1 x + \alpha_0)(b_2x^2 + b_1x + b_0)$$

$$\begin{array}{lllllllll} x^3: & \beta_1 a_2 & & & = & \alpha_1 b_2 & & \\ x^2: & \beta_1 a_1 & + & \beta_0 a_2 & = & \alpha_1 b_1 & + & \alpha_0 b_2 \\ x^1: & \beta_1 a_0 & + & \beta_0 a_1 & = & \alpha_1 b_0 & + & \alpha_0 b_1 \\ x^0: & & & \beta_0 a_0 & = & & & \alpha_0 b_0 \end{array}$$

Resultants as Determinants

We solve a linear system in β_1, β_0, α_1, and α_0:

$$\left\{ \begin{array}{lcrcrcr} \beta_1 a_2 & & & = & \alpha_1 b_2 & & \\ \beta_1 a_1 & + & \beta_0 a_2 & = & \alpha_1 b_1 & + & \alpha_0 b_2 \\ \beta_1 a_0 & + & \beta_0 a_1 & = & \alpha_1 b_0 & + & \alpha_0 b_1 \\ & & \beta_0 a_0 & = & & & \alpha_0 b_0 \end{array} \right.$$

In matrix form:

$$\left[\begin{array}{cccc} a_2 & 0 & b_2 & 0 \\ a_1 & a_2 & b_1 & b_2 \\ a_0 & a_1 & b_0 & b_1 \\ 0 & a_0 & 0 & b_0 \end{array} \right] \left[\begin{array}{c} \beta_1 \\ \beta_0 \\ -\alpha_1 \\ -\alpha_0 \end{array} \right] = \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \end{array} \right]$$

Condition for nonzero solution: determinant of matrix is zero.

Discriminants as Resultants

Solving the quadratic equation: $ax^2 + bx + c = 0$...

```
sage: R.<x,a,b,c> = QQ[]
sage: p = a*x^2 + b*x + c
sage: dp = diff(p,x)
sage: disc = singular.resultant(p,dp,x)
sage: print disc
sage: print factor(R(disc))
```

The discriminant is a polynomial in the coefficients and vanishes whenever the polynomial and its derivative have a common solution. The output:

```
-a*b^2+4*a^2*c
 a * (-b^2 + 4*a*c)
```

Computing Resultants

We first declare a polynomial ring with rational coefficients

```
sage: R.<x,y,c,r> = QQ[]
sage: F = R(f)
sage: G = R(g)
sage: D = R(dJ)
```

We can now eliminate x using the resultant from `Singular`.

```
sage: rFG = singular.resultant(F,G,x)
sage: print rFG
sage: rGD = singular.resultant(G,D,x)
sage: print rGD
```

The result:

```
4*y^2*c^2+c^4-2*c^2*r^2+r^4-2*c^2-2*r^2+1
16*y^2*c^2
```

The Discriminant of our Circle Problem

an algebraic condition on the parameters of the problem

```
sage: discriminant = singular.resultant(rFG,rGD,y)
sage: print discriminant

256*c^12-1024*c^10*r^2+1536*c^8*r^4-1024*c^6*r^6
+256*c^4*r^8-1024*c^10+1024*c^8*r^2+1024*c^6*r^4
-1024*c^4*r^6+1536*c^8+1024*c^6*r^2+1536*c^4*r^4
-1024*c^6-1024*c^4*r^2+256*c^4
```

Geometric interpretation:
$\rightarrow$ the discriminant gives the relation between center $(c, 0)$ and radius r of the second circle for which the solutions are singular, i.e.:

1. double solutions: circles touch each other
2. a solution set: overlapping circles

Factoring the Discriminant

to simplify the condition on the parameters

To factor the discriminant, we must convert to an element of the ring R.

```
sage: print type(discriminant)
sage: factor(R(discriminant))

<class 'sage.interfaces.singular.SingularElement'>
(256) * (c - r - 1)^2 * (c - r + 1)^2
      * (c + r - 1)^2 * (c + r + 1)^2 * c^4
```

So the discriminant for our problem looks as follows:

$$256(c-r-1)^2(c-r+1)^2(c+r-1)^2(c+r+1)^2c^4$$

Collecting all Formulas

The system

$$\left\{ \begin{array}{r} x^2 + y^2 - 1 = 0 \\ (x - c)^2 + y^2 - r^2 = 0 \end{array} \right.$$

has exactly two solutions

$$\left[x = \frac{-r^2 + c^2 + 1}{2c}, y = \pm \frac{\sqrt{-r^4 + 2c^2r^2 + 2r^2 - c^4 + 2c^2 - 1}}{2c} \right]$$

except for those c and r satisfying

$$256(c - r - 1)^2(c - r + 1)^2(c + r - 1)^2(c + r + 1)^2c^4 = 0.$$

The Discriminant Variety

plot of $256(c-r-1)^2(c-r+1)^2(c+r-1)^2(c+r+1)^2c^4 = 0$



Considering only the positive values for c and r,
we classify the regular solutions in four different configurations.

Suggested Explorations

Some variations of the problem we considered:

1. Replace the second circle by a general ellipse.
2. Use a polynomial of degree three in the second equation.
3. Consider the problem of intersecting two ellipses.
4. Examine the intersection of three spheres.